

# CALCULATIONS INVOLVING SYMBOLIC POWERS

BEN DRABKIN, ELOÍSA GRIFO, ALEXANDRA SECELEANU, AND BRANDEN STONE

ABSTRACT. Symbolic powers are a classical commutative algebra topic that relates to primary decomposition, consisting, in some circumstances, of the functions that vanish up to a certain order on a given variety. However, these are notoriously difficult to compute, and there are seemingly simple questions related to symbolic powers that remain open even over polynomial rings. In this paper, we describe a *Macaulay2* software package that allows for computations of symbolic powers of ideals and which can be used to study the equality and containment problems, among others.

## 1. INTRODUCTION

Given an ideal  $I$  in a Noetherian domain  $R$ , the  $n$ -th *symbolic power* of  $I$  is the ideal defined by

$$I^{(n)} = \bigcap_{P \in \text{Ass}(I)} (I^n R_P \cap R).$$

When  $I$  itself has no embedded primes, the minimal primes of  $I^n$  coincide with the associated primes of  $I$ , and  $I^{(n)}$  as above corresponds to the intersection of the primary components corresponding to minimal primes of  $I^n$ . In particular, under these circumstances the definition is unchanged if instead we have  $P$  ranging over  $\text{Min}(I)$ . However, if we consider any ideal  $I$ , with no assumptions on its associated primes, there are two possible definitions of symbolic powers: the one above, and the one given by

$$I^{(n)} = \bigcap_{P \in \text{Min}(I)} (I^n R_P \cap R).$$

The `SymbolicPowers` package, in <https://github.com/eloisagrifo/SymbolicPowers>, allows the user to compute the symbolic powers of any ideal over a polynomial ring, taking the first definition as the standard, but allowing the user to take the second definition instead via the option `UseMinimalPrimes`, which can be used in any method.

Symbolic powers are a classical topic that relates to many subjects within commutative algebra and algebraic geometry, and an active area of current research. If  $P$  is a prime ideal in a polynomial ring, the classical Zariski–Nagata [Zar49, Nag62] theorem says that the symbolic powers of  $P$  consist of the functions that vanish up to order  $n$  in the corresponding variety. Over a perfect field, these coincide with differential powers. For a survey on symbolic powers, see [DDSG<sup>+</sup>17].

Various invariants have been defined to compare symbolic and ordinary powers of ideals: the resurgence [BH10], the Waldschmidt constant [BH10], and symbolic defect [GGST16], among others. These can be in some cases explicitly computed and in others approximated using the `SymbolicPowers` package.

---

2010 *Mathematics Subject Classification*. Primary 13P99. Secondary: 13A15, 13C99.

*Key words and phrases*. symbolic powers, Macaulay2.

## 2. BASIC USAGE

The main method in the `SymbolicPowers` package is `symbolicPower`, which given an ideal  $I$  and an integer  $n$  returns  $I^{(n)}$ . Computations are done using the standard definition of symbolic powers; if the option `UseMinimalPrimes` is set true, then the definition of symbolic powers used in the computations will be the non-standard one, as described in the introduction. When `UseMinimalPrimes` is set true, the algorithm takes a primary decomposition of  $I^n$  and intersects the components corresponding to minimal primes. Through the rest of the paper, we will assume that the `UseMinimalPrimes` option is set to false, which is the default setting.

The fastest algorithm used in `symbolicPower` is the one for homogeneous ideals  $I$  of height  $\dim(R) - 1$  in  $R$ . In this case, the  $I^{(n)}$  coincides with the saturation of  $I^n$  with respect to the maximal ideal, as can be seen in the following example.

**Example 2.1** (Height of ideal is  $\dim(R) - 1$ ).

```
i1 : loadPackage "SymbolicPowers"

i2 : R=QQ[x,y,z];

i3 : I=ideal(x*(y^3-z^3),y*(z^3-x^3),z*(x^3-y^3));
o3 : ideal of R

i4 : symbolicPower(I,3);
o4 : ideal of R

i4 : symbolicPower(I,3)==saturate(I^3)
o4 : True
```

If  $I$  is a primary ideal, a primary decomposition of  $I^n$  is scanned for the component with radical  $I$ .

**Example 2.2** (Primary ideals).

```
i1 : loadPackage "SymbolicPowers";

i2 : R=QQ[w,x,y,z]/(x*y-z^2);

i3 : I=ideal(x,z);
o3 : Ideal of R

i4 : symbolicPower(I,2)
o4 = ideal x
o4 : Ideal of R
```

If  $I$  is a monomial ideal, the symbolic powers are computed by intersecting the powers of the associated primes in the squarefree case, or by explicitly picking out the appropriate components of the primary decomposition of  $I^n$  in the general case. This does not require that  $I$  be of type `monomialIdeal`, but it does make use of the fact that the `primaryDecomposition` algorithm is faster for ideals of type `monomialIdeal`.

**Example 2.3** (Monomial ideals).

```

i1 : loadPackage "SymbolicPowers";

i2 : R = QQ[x,y,z];

i3 : I = ideal(x*y,x*z,y*z)
o3 = ideal (x*y, x*z, y*z)
o3 : Ideal of QQ[x, y, z]

i4 : symbolicPower(I,2)
      2 2    2 2    2 2
o4 = ideal (x*y*z, y z , x z , x y )
o4 : Ideal of QQ[x, y, z]

```

### 3. APPLICATIONS

**3.1. Equality.** Symbolic powers do not, in general, coincide with the ordinary powers, even in the case of prime ideals. In fact, the question of characterizing the ideals  $I$  for which  $I^{(n)} = I^n$  for all  $n$  is essentially open. Using `isSymbolicEqualOrdinary`, one can determine if the  $n$ -th symbolic and ordinary powers of a given ideal coincide, often without computing the actual symbolic power of  $I$ . For this, the package makes use of `bigHeight`.

The method `bigHeight` computes the largest height of an associated prime of  $I$ . Similarly, `minimalPart` returns the intersection of the minimal components of a given ideal. Instead of explicitly finding the associated primes of  $I$  and taking their heights, the following result is used [EHV92].

**Theorem 3.1** (Eisenbud–Huneke–Vasconcellos, 1992). Given an ideal  $I$  in a regular ring  $R$  of height  $h$ , then for each  $e \geq h$ ,  $I$  has an associated prime of height  $e$  if and only if the height of  $\text{Ext}^e(R/I, R)$  is  $e$ .

This is faster than computing the set of associated primes. To determine if  $I^{(n)} = I^n$  for a specific value of  $n$ , `isSymbolicEqualOrdinary` first compares the big heights of  $I^n$  and  $I$ : if the big heights differ, then  $I^n$  must have embedded components, and `isSymbolicEqualOrdinary` returns `false`; if the big heights are both equal to the height of  $I$ , then  $I^n$  cannot have embedded components, and `isSymbolicEqualOrdinary` returns `true`.

**3.2. The Containment Problem.** The Containment Problem for ordinary and symbolic powers of ideals consists of answering the following question: given an ideal  $I$ , for which values of  $a$  and  $b$  does the containment  $I^{(a)} \subseteq I^b$  hold? Over a regular ring, a well-known theorem of Ein–Lazarsfeld–Smith, Hochster–Huneke and Ma–Schwede [ELS01, HH02, MS17] gives a partial answer to this question: when  $I$  is a radical ideal,  $I^{(hn)} \subseteq I^n$  holds for all  $n$ , where  $h$  denotes the big height of the ideal  $I$ . However, this is not necessarily best possible. Using `containmentProblem`, the user can either determine the smallest value of  $a$  given  $b$  for which  $I^{(a)} \subseteq I^b$ , or the largest value of  $b$  given  $a$  for which the same containment holds.

**Example 3.2** (Containment Problem).

```

i1 : loadPackage "SymbolicPowers";

i2 : R=QQ[x,y,z];

```

```

i3 : I=ideal(x*(y^3-z^3),y*(z^3-x^3),z*(x^3-y^3));
o3 : ideal of R

i4 : containmentProblem(I,2)
o4 = 4

```

#### 4. ASYMPTOTIC INVARIANTS

In an effort to make progress on the containment problem, various asymptotic interpolation invariants have been proposed by Bocci and Harbourne [BH10]. One such invariant is the Waldschmidt constant for a homogeneous ideal  $I$ . This is an asymptotic measure of the initial degree of the symbolic powers of  $I$ . The *initial degree* of a homogeneous ideal  $I$  is  $\alpha(I) = \min\{d \mid I_d \neq 0\}$ , i.e. the smallest degree of a nonzero element in  $I$ . The *Waldschmidt constant* of  $I$  is defined to be

$$\hat{\alpha}(I) = \lim_{m \rightarrow \infty} \frac{\alpha(I^{(m)})}{m}.$$

Due to the asymptotic nature of the Waldschmidt constant, there is no a priori algorithm to determine this for arbitrary ideals. An important exception is the case when the ideal  $I$  is a monomial ideal. In this context, the Waldschmidt constant can be computed as the smallest among the sums of the coordinates of all points in a convex body termed the *symbolic polyhedron* of  $I$  [SMCH, BCG<sup>+</sup>16]. Our package computes Waldschmidt constants of monomial ideals by finding their symbolic polyhedron. The `symbolicPolyhedron` routine makes heavy use of the *Polyhedra* package by René Birkner, which in turn relies on the *FourierMotzkin* package by Greg Smith. This allows to determine the Waldschmidt constants of monomial ideals exactly as in the following example.

**Example 4.1** (Waldschmidt constant of monomial ideals).

```

i1 : loadPackage "SymbolicPowers";

i2 : R=QQ[x,y,z];

i3 : I=ideal(x*y,x*z,y*z);
o3 : Ideal of R

i4 : symbolicPolyhedron(I)
o4 = {ambient dimension => 3
      dimension of lineality space => 0
      dimension of polyhedron => 3
      number of facets => 6
      number of rays => 3
      number of vertices => 4
o4 : Polyhedron

i5 : waldschmidt I
Ideal is monomial, the Waldschmidt constant is computed exactly

```

```

      3
o5 = -
      2
o5 : QQ

```

In the case of arbitrary ideals, the Waldschmidt constant is approximated by taking the minimum of the values  $\frac{\alpha(I^{(m)})}{m}$ , where  $m$  ranges from 1 to a specified optional input `SampleSize`.

**Example 4.2** (Waldschmidt constant of arbitrary ideals).

```

i1 : loadPackage "SymbolicPowers";

i2 : R=QQ[x,y,z];

i3 : I=ideal(x*(y^3-z^3),y*(z^3-x^3),z*(x^3-y^3));
o3 : Ideal of R

i4 : waldschmidt I
Ideal is not monomial, the Waldschmidt constant is approximated
using first 10 powers.
o4 = 3
o4 : QQ

```

Note that the true value for the Waldschmidt constant of the above ideal is indeed 3 as proven in [DHN<sup>+</sup>15]. In general, for an ideal that is not monomial, the function `waldschmidt` will return an upper bound on the true value of the Waldschmidt constant.

Another asymptotic invariant termed *resurgence* is defined as

$$\rho(I) = \sup \left\{ \frac{m}{r} \mid I^{(m)} \not\subseteq I^r \right\}.$$

There are no algorithms known to date that compute resurgence exactly, therefore our package computes a lower bound for the resurgence by taking the maximum of the values  $\frac{m}{r}$ , where  $r$  ranges from 1 to the second input of the function `lowerBoundResurgence`. Continuing with the ideal in the previous example, we compute a lower bound on its resurgence.

**Example 4.3** (Lower bound on resurgence).

```

i5 : lowerBoundResurgence(I,5)
      3
o5 = -
      2
o5 : QQ

```

Note that the value for the resurgence of the ideal in this example is also  $\frac{3}{2}$  by [DHN<sup>+</sup>15].

#### ACKNOWLEDGMENTS

We would like to thank the organizers of the July 2017 Macaulay2 Workshop at the University of California – Berkeley, where a large portion of this work was done. The code for computing the symbolic polyhedron and Waldschmidt constant of a monomial ideal was developed by the third named author in collaboration with Andrew Conner and Xuehua (Diana) Zhong. We thank them for their contribution to these routines.

## REFERENCES

- [BCG<sup>+</sup>16] Cristiano Bocci, Susan Cooper, Elena Guardo, Brian Harbourne, Mike Janssen, Uwe Nagel, Alexandra Seceleanu, Adam Van Tuyl, and Thanh Vu. The Waldschmidt constant for squarefree monomial ideals. *J. Algebraic Combin.*, 44(4):875–904, 2016.
- [BH10] Cristiano Bocci and Brian Harbourne. Comparing powers and symbolic powers of ideals. *J. Algebraic Geom.*, 19(3):399–417, 2010.
- [DDSG<sup>+</sup>17] Hailong Dao, Alessandro De Stefani, Eloísa Grifo, Craig Huneke, and Luis Núñez-Betancourt. *Symbolic powers of ideals*. Springer Proceedings in Mathematics & Statistics. Springer, 2017.
- [DHN<sup>+</sup>15] M. Dumnicki, B. Harbourne, U. Nagel, A. Seceleanu, T. Szemberg, and H. Tutaj-Gasińska. Resurgences for ideals of special point configurations in  $\mathbf{P}^N$  coming from hyperplane arrangements. *J. Algebra*, 443:383–394, 2015.
- [DSTG13] Marcin Dumnicki, Tomasz Szemberg, and Halszka Tutaj-Gasińska. Counterexamples to the  $I^{(3)} \subset I^2$  containment. *J. Algebra*, 393:24–29, 2013.
- [EHV92] David Eisenbud, Craig Huneke, and Wolmer Vasconcelos. Direct methods for primary decomposition. *Inventiones mathematicae*, 110(1):207–235, 1992.
- [ELS01] Lawrence Ein, Robert Lazarsfeld, and Karen E. Smith. Uniform bounds and symbolic powers on smooth varieties. *Invent. Math.*, 144(2):241–252, 2001.
- [GGST16] Federico Galetto, Anthony V. Geramita, Yong-Su Shin, and Adam Van Tuyl. The symbolic defect of an ideal, 2016.
- [GVV05] Isidoro Gitler, Carlos Valencia, and Rafael H. Villarreal. A note on the Rees algebra of a bipartite graph. *J. Pure Appl. Algebra*, 201(1-3):17–24, 2005.
- [HH02] Melvin Hochster and Craig Huneke. Comparison of symbolic and ordinary powers of ideals. *Invent. Math.* 147 (2002), no. 2, 349–369, November 2002.
- [MS17] Linqun Ma and Karl Schwede. Perfectoid multiplier/test ideals in regular rings and bounds on symbolic powers. *arXiv:1705.02300*, 2017.
- [Nag62] Masayoshi Nagata. *Local rings*. Interscience, 1962.
- [SMCH] Tai Ha Susan M. Cooper, Robert J. D. Embree and Andrew H. Hoefel. Symbolic powers of monomial ideals.
- [Zar49] Oscar Zariski. A fundamental lemma from the theory of holomorphic functions on an algebraic variety. *Ann. Mat. Pura Appl. (4)*, 29:187–198, 1949.

DEPARTMENT OF MATHEMATICS, UNIVERSITY OF NEBRASKA, LINCOLN, NE 68588-0130, USA  
*E-mail address:* benjamin.drabkin@huskers.unl.edu

DEPARTMENT OF MATHEMATICS, UNIVERSITY OF VIRGINIA, CHARLOTTESVILLE, VA 22904-4135, USA  
*E-mail address:* eloisa.grifo@virginia.edu

DEPARTMENT OF MATHEMATICS, UNIVERSITY OF NEBRASKA, LINCOLN, NE 68588-0130, USA  
*E-mail address:* aseceleanu@unl.edu

DEPT OF MATH AND COMPUTER SCIENCE, ADELPHI UNIVERSITY, GARDEN CITY, NY 11530-0701, USA  
*E-mail address:* bstone@adelphi.edu